Computergestuetzte Mathematik zur Analysis

Lektion 8

```
> restart:
```

Einige bestimmte Integrale

```
> A := Int(exp(-x^2), x = -infinity .. infinity);
```

$$A := \int_{-\infty}^{\infty} e^{-x^2} \, dx \tag{1.1}$$

```
> value(A);
```

$$\sqrt{\pi} \tag{1.2}$$

```
> B := Int(exp(-x^2), x);
```

$$B := \int e^{-x^2} \, dx \tag{1.3}$$

```
> value(B);
```

$$\frac{1}{2} \sqrt{\pi} \operatorname{erf}(x) \tag{1.4}$$

```
> C := Int(ln(x)^2 / (1 + x)^2, x);
```

$$C := \int \frac{\ln(x)^2}{(1+x)^2} \, dx \tag{1.5}$$

```
> value(C);
```

$$\int \frac{\ln(x)^2}{(1+x)^2} \, dx \tag{1.6}$$

```
> E := Int(ln(x)^2 / (1 + x)^2, x = 0 .. infinity);
```

$$E := \int_{0}^{\infty} \frac{\ln(x)^2}{(1+x)^2} \, dx \tag{1.7}$$

```
> value(E);
```

$$\frac{1}{3} \pi^2 \tag{1.8}$$

```
> evalf(E, 15);
```

$$\int_{0.}^{Float(\infty)} \frac{\ln(x)^2}{(1.+x)^2} \, dx \tag{1.9}$$

```
> evalf(value(E), 15);
```

(1.10)

$$3.28986813369644 \quad (1.10)$$

Riemann Integral

```
> restart:
> S := Sum(a/n*exp(k*a/n), k = 0 .. n-1);
```

$$S := \sum_{k=0}^{n-1} \frac{a\,e^{\frac{k\,a}{n}}}{n} \quad (2.1)$$

```
> value(S);
```

$$\frac{a\,e^{a}}{\left(e^{\frac{a}{n}}-1\right)n} - \frac{a}{\left(e^{\frac{a}{n}}-1\right)n} \quad (2.2)$$

```
> L := Limit(S, n = infinity);
```

$$L := \lim_{n\to\infty} \left(\sum_{k=0}^{n-1} \frac{a\,e^{\frac{k\,a}{n}}}{n}\right) \quad (2.3)$$

```
> value(L);
```

$$e^{a}-1 \quad (2.4)$$

Die Gamma-Funktion

```
> restart:
> g := t^(x-1)*exp(-t);
```

$$g := t^{x-1}\,e^{-t} \quad (3.1)$$

```
> G := Int(g, t = 0 .. infinity);
```

$$G := \int_0^{\infty} t^{x-1}\,e^{-t}\,dt \quad (3.2)$$

```
> value(G);
```

$$\Gamma(x) \quad (3.3)$$

```
> GAMMA(x+1) = expand(GAMMA(x+1));
```

$$\Gamma(x+1) = \Gamma(x)\,x \quad (3.4)$$

induktiv zeigt man, dass Gamm(n+1) = n!

```
> GAMMA(1/2); GAMMA(1/2)/(-1/2); GAMMA(-1/2);
```

$$\sqrt{\pi}$$

$$-2\sqrt{\pi}$$

$$-2\sqrt{\pi} \quad (3.5)$$

```
> plot(GAMMA(x),x=-2..5,-25..25,discont=true,thickness=2);
```




```
> a:=x*exp(gamma*x)*Product( (1+x/n)*exp(-x/n),n=1..k);
```

$$a := x\,e^{\gamma x} \prod_{n=1}^{k} \left(\left(1 + \frac{x}{n}\right) e^{-\frac{x}{n}} \right) \qquad (3.6)$$

```
> value(Limit(a, k = infinity));
```

$$\frac{1}{\Gamma(x)} \qquad (3.7)$$

Grenzwerte

```
> restart:
> a := k^2/(2^k);
```

$$a := \frac{k^2}{2^k} \qquad (4.1)$$

```
> A := Limit(a, k = infinity);
```

$$A := \lim_{k \to \infty} \frac{k^2}{2^k} \qquad (4.2)$$

```
> value(A);
```
$$\lim_{k\to\infty} \frac{k^2}{2^k} \qquad (4.3)$$

```
> b := k^k/k!;
```
$$b := \frac{k^k}{k!} \qquad (4.4)$$

```
> B := Limit(b, k = infinity);
```
$$B := \lim_{k\to\infty} \frac{k^k}{k!} \qquad (4.5)$$

```
> value(B);
```
$$\infty \qquad (4.6)$$

```
> c := (-1)^k*b;
```
$$c := \frac{(-1)^k k^k}{k!} \qquad (4.7)$$

```
> C := Limit(c, k = infinity);
```
$$C := \lim_{k\to\infty} \frac{(-1)^k k^k}{k!} \qquad (4.8)$$

```
> value(C);
```
$$\mathit{undefined} \qquad (4.9)$$

```
> e := sin(k*Pi);
```
$$e := \sin(k\pi) \qquad (4.10)$$

```
> E := Limit(e, k = infinity);
```
$$E := \lim_{k\to\infty} \sin(k\pi) \qquad (4.11)$$

```
> value(E);
```
$$-1 .. 1 \qquad (4.12)$$

```
> value(E) assuming k::integer;
```
$$-1 .. 1 \qquad (4.13)$$

```
> e assuming k::integer;
```
$$0 \qquad (4.14)$$

Reihen

```
> restart:
> a := 1/k/(k+1);
```
$$a := \frac{1}{k(k+1)} \qquad (5.1)$$

```
> A := Sum(a, k = 1 .. infinity);
```
$$A := \sum_{k=1}^{\infty} \frac{1}{k(k+1)} \qquad (5.2)$$

```
> value(A);
```

$$1 \qquad (5.3)$$

```
> A1 := Sum(a, k = 1 .. N);
```

$$A1 := \sum_{k=1}^{N} \frac{1}{k\,(k+1)} \qquad (5.4)$$

```
> value(A1);
```

$$-\frac{1}{N+1} + 1 \qquad (5.5)$$

```
> b := 1/k^3;
```

$$b := \frac{1}{k^3} \qquad (5.6)$$

```
> B := Sum(b, k = 1 .. infinity);
```

$$B := \sum_{k=1}^{\infty} \frac{1}{k^3} \qquad (5.7)$$

```
> value(B);
```

$$\zeta(3) \qquad (5.8)$$

```
> evalf(%);
```

$$1.202056903 \qquad (5.9)$$

```
> product((1-z^j)/(1+z^j),j=1..infinity);
product: Cannot show that (1-z^j)/(1+z^j) is continuous on
[1,infinity]
```

$$\prod_{j=1}^{\infty} \left(\frac{1-z^j}{1+z^j} \right) \qquad (5.10)$$

Gleichmaessige Konvergenz

```
> a := (4*sin(x)*(1/5))^k;
```

$$a := \left(\frac{4}{5} \sin(x) \right)^k \qquad (6.1)$$

```
> S := Sum(a, k = 1 .. infinity);
```

$$S := \sum_{k=1}^{\infty} \left(\frac{4}{5} \sin(x) \right)^k \qquad (6.2)$$

```
> f := value(S);
```

$$f := -\frac{4\sin(x)}{4\sin(x) - 5} \qquad (6.3)$$

```
> farben := [black, red, yellow, blue, green, magenta, coral,
pink, cyan];
```

$$farben := [black, red, yellow, blue, green, magenta, coral, pink, cyan] \qquad (6.4)$$

```
> funktionen := [f, seq(sum(a, k = 1 .. n), n = 1 .. 7)];
```

$$funktionen := \left[-\frac{4\sin(x)}{4\sin(x) - 5}, \frac{4}{5}\sin(x), \frac{4}{5}\sin(x) + \frac{16}{25}\sin(x)^2, \frac{4}{5}\sin(x) \right. \qquad (6.5)$$

$$+ \frac{16}{25}\sin(x)^2 + \frac{64}{125}\sin(x)^3, \frac{4}{5}\sin(x) + \frac{16}{25}\sin(x)^2 + \frac{64}{125}\sin(x)^3 + \frac{256}{625}\sin(x)^4, \frac{4}{5}\sin(x) + \frac{16}{25}\sin(x)^2 + \frac{64}{125}\sin(x)^3 + \frac{256}{625}\sin(x)^4 + \frac{1024}{3125}\sin(x)^5, \frac{4}{5}\sin(x) + \frac{16}{25}\sin(x)^2 + \frac{64}{125}\sin(x)^3 + \frac{256}{625}\sin(x)^4 + \frac{1024}{3125}\sin(x)^5 + \frac{4096}{15625}\sin(x)^6, \frac{4}{5}\sin(x) + \frac{16}{25}\sin(x)^2 + \frac{64}{125}\sin(x)^3 + \frac{256}{625}\sin(x)^4 + \frac{1024}{3125}\sin(x)^5 + \frac{4096}{15625}\sin(x)^6 + \frac{16384}{78125}\sin(x)^7\Big]$$

```
> plot(funktionen, x = -Pi .. Pi, color = farben);
```




```
> funktionen := [f, f+1/3, f-1/3, seq(sum(a, k = 1 .. 5*n), n =
  1 .. 4)]:
> farben := [black, gray, gray, red, yellow, blue, green,
  magenta, coral, pink, cyan];
```

farben := [*black*, *gray*, *gray*, *red*, *yellow*, *blue*, *green*, *magenta*, *coral*, *pink*, *cyan*] **(6.6)**

```
> plot(funktionen, x = -Pi .. Pi, color = farben, thickness =
  2, numpoints = 500);
```



Das Taylorpolynom

```
> f := sqrt(1+x)/sqrt(1+x^2);
```

$$f := \frac{\sqrt{1+x}}{\sqrt{1+x^2}} \tag{7.1}$$

```
> t := series(f, x  = 0, 8);
```

$$t := 1 + \frac{1}{2}\,x - \frac{5}{8}\,x^2 - \frac{3}{16}\,x^3 + \frac{51}{128}\,x^4 + \frac{47}{256}\,x^5 - \frac{369}{1024}\,x^6 - \frac{267}{2048}\,x^7 + \mathrm{O}(x^8) \tag{7.2}$$

```
> P := convert(t, polynom);
```

$$P := 1 + \frac{1}{2}\,x - \frac{5}{8}\,x^2 - \frac{3}{16}\,x^3 + \frac{51}{128}\,x^4 + \frac{47}{256}\,x^5 - \frac{369}{1024}\,x^6 - \frac{267}{2048}\,x^7 \tag{7.3}$$

```
> for n from 1 to 3 do;
>   t := series(f, x = 0, n + 1);
```

```
>   P[n] := convert(t, polynom);
> od;
```

$$t := 1 + \frac{1}{2}\,x + \mathrm{O}(x^2)$$

$$P_1 := 1 + \frac{1}{2}\,x$$

$$t := 1 + \frac{1}{2}\,x - \frac{5}{8}\,x^2 + \mathrm{O}(x^3)$$

$$P_2 := 1 + \frac{1}{2}\,x - \frac{5}{8}\,x^2$$

$$t := 1 + \frac{1}{2}\,x - \frac{5}{8}\,x^2 - \frac{3}{16}\,x^3 + \mathrm{O}(x^4)$$

$$P_3 := 1 + \frac{1}{2}\,x - \frac{5}{8}\,x^2 - \frac{3}{16}\,x^3 \quad \text{(7.4)}$$

```
> P[0] := f;
```

$$P_0 := \frac{\sqrt{1+x}}{\sqrt{1+x^2}} \quad \text{(7.5)}$$

```
> farbe := [black,blue, red, cyan];
```

$$farbe := [black, blue, red, cyan] \quad \text{(7.6)}$$

```
> plot(convert(P, list), x = -.8 .. .8, color = farbe,
  thickness = 2);
```




```
> g := cos(x);
```

$$g := \cos(x) \quad (7.7)$$

```
> for n in [4, 20, 60] do;
>   Q[n] := convert(series(g, x, n+1), polynom):
    E[n] := Q[n] - g;
> od:
> Q[4];
```

$$1 - \frac{1}{2} x^2 + \frac{1}{24} x^4 \quad (7.8)$$

> $E[0] := 0;$

$$E_0 := 0 \quad (7.9)$$

```
> Q[0] := g;
```

$$Q_0 := \cos(x) \quad (7.10)$$

```
> plot(convert(Q, list), x = - 32 .. 32, y = -2 .. 2, color =
  farbe, thickness = 2, numpoints = 1000);
```



Das komplexe Bild

```
> restart:
> h := arctan(x);
```

$$h := \arctan(x) \quad (8.1)$$

```
> for n in [4, 20, 60] do;
>   S[n] := convert(series(h, x = 0, n+1), polynom):
> od;
```

$$S_4 := x - \frac{1}{3}x^3$$

$$S_{20} := x - \frac{1}{3}x^3 + \frac{1}{5}x^5 - \frac{1}{7}x^7 + \frac{1}{9}x^9 - \frac{1}{11}x^{11} + \frac{1}{13}x^{13} - \frac{1}{15}x^{15} + \frac{1}{17}x^{17} - \frac{1}{19}x^{19}$$

$$S_{60} := x + \frac{1}{5}x^5 - \frac{1}{3}x^3 + \frac{1}{21}x^{21} - \frac{1}{7}x^7 + \frac{1}{9}x^9 - \frac{1}{11}x^{11} + \frac{1}{13}x^{13} - \frac{1}{15}x^{15} \quad (8.2)$$

$$+\frac{1}{17}x^{17}-\frac{1}{19}x^{19}-\frac{1}{23}x^{23}+\frac{1}{25}x^{25}-\frac{1}{27}x^{27}+\frac{1}{29}x^{29}-\frac{1}{31}x^{31}$$
$$+\frac{1}{33}x^{33}-\frac{1}{35}x^{35}+\frac{1}{37}x^{37}-\frac{1}{39}x^{39}+\frac{1}{41}x^{41}-\frac{1}{43}x^{43}+\frac{1}{45}x^{45}$$
$$-\frac{1}{47}x^{47}+\frac{1}{49}x^{49}-\frac{1}{51}x^{51}+\frac{1}{53}x^{53}-\frac{1}{55}x^{55}+\frac{1}{57}x^{57}-\frac{1}{59}x^{59}$$

```
> S[0] := h;
```

$$S_0 := \arctan(x) \quad \textbf{(8.3)}$$

```
> plot(convert(S, list), x = -2 .. 2, -2 .. 2, thickness = 2,
  numpoints = 500);
```




```
> plot([h, S[60]], x = -2 .. 2, -2 .. 2, color = [blue, green],
  thickness = 2, numpoints = 500);
```




```
> x := a + I*b;
```

$$x := a + \mathrm{I}\, b \tag{8.4}$$

```
> pl1 := plot3d(abs(h-S[20]), a = -1.3 .. 1.3, b = -1.3 .. 1.3,
  shading = zhue,  style = patchcontour, numpoints = 2000):
> with(plots):
> display(pl1, axes = boxed, view = 0 .. 5, orientation =
  [-110, 35]);
```




```
> pl2 := plot3d(abs(h), a = -1.3 .. 1.3, b = -1.3 .. 1.3, color
  = black,  style = wireframe, numpoints = 4000):
> display([pl1, pl2], axes = boxed, view = 0 .. 2, orientation
  = [-110, 35]);
```




```
> x := 'x';
```

$$x := x \tag{8.5}$$

```
> plot(abs(arctan(I*x)),x=-2..2);
```



allgemeinere Reihenentwicklungen

```
> pl1 := plot([h, S[60]], x = -2 .. 3, -2 .. 2, color = [blue,
  green], thickness = 2, numpoints = 500):
> pl1;
```




```
> series(h, x = infinity, 12);
```

$$\frac{1}{2}\pi - \frac{1}{x} + \frac{1}{3x^3} - \frac{1}{5x^5} + \frac{1}{7x^7} - \frac{1}{9x^9} + \frac{1}{11x^{11}} + O\left(\frac{1}{x^{12}}\right) \quad (9.1)$$

```
> pl2 := plot(convert(%, polynom), x = .1 .. 3, -2 .. 2,
  thickness = 2):
> display({pl1, pl2});
```




```
> series(h, x = 1, 26):
> pl3 := plot(convert(%, polynom), x = -2 .. 3, -2 .. 2, color
  = black, thickness = 2):
> display({pl1, pl2, pl3});
```




```
> a := ln(x)^2/(1+x)^2;
```

$$a := \frac{\ln(x)^2}{(x+1)^2} \quad (9.2)$$

```
> series(a, x = infinity);
```

$$\frac{\ln(x)^2}{x^2} - \frac{2\ln(x)^2}{x^3} + \frac{3\ln(x)^2}{x^4} - \frac{4\ln(x)^2}{x^5} + O\left(\frac{1}{x^6}\right) \quad (9.3)$$

```
> f := 1 - cos(x^2);
```

$$f := 1 - \cos(x^2) \quad (9.4)$$

```
> g := x*(x - sin(x));
```

$$g := x\,(x - \sin(x)) \quad (9.5)$$

```
> series(f, x, 10) / series(g, x, 10);
```

$$\frac{\frac{1}{2}x^4 - \frac{1}{24}x^8 + O(x^{10})}{\frac{1}{6}x^4 - \frac{1}{120}x^6 + \frac{1}{5040}x^8 + O(x^{10})} \quad (9.6)$$

```
> b := convert(%, polynom);
```

$$b := \frac{\frac{1}{2}x^4 - \frac{1}{24}x^8}{\frac{1}{6}x^4 - \frac{1}{120}x^6 + \frac{1}{5040}x^8} \tag{9.7}$$

```
> normal(b, expanded);
```

$$\frac{2520 - 210\,x^4}{840 - 42\,x^2 + x^4} \tag{9.8}$$

```
> subs(x = 0, %);
```

$$3 \tag{9.9}$$

```
> limit(f/g, x = 0);
```

$$3 \tag{9.10}$$